

第3章 重回帰分析

【課題】教科書 108-109 ページの例 5 のデータ（表 3-3）について、

- ① TSP を用いて重回帰分析をおこなってみる。
 - ② 相関係数行列と偏相関係数行列を求める。
- という 2 つのことをおこなってみよう。

1. Excelデータの作成

分析の準備として、Excel のデータファイルを作成しておく。ここでは、**ex3-3.csv** という名前で、**CSV 形式**で保存する。

	A	B	C	D
1	1983	185234	219047	214658.4
2	1984	189610.5	223793.4	219047
3	1985	197124.2	230667.3	223793.4
4	1986	203577.3	236083.4	230667.3
5	1987	211994.4	240775.4	236083.4
6	1988	222311.8	253954.6	240775.4
7	1989	233017.5	265997.6	253954.6
8	1990	243711.2	279366.1	265997.6
9	1991	250592.4	291102.9	279366.1
10	1992	256900.5	295572.9	291102.9
11	1993	260282.1	297769.1	295572.9
12	1994	267276	302224.7	297769.1
13	1995	271955.8	304958.7	302224.7
14	1996	278575.1	305548.7	304958.7
15	1997	281151.7	309193	305548.7
16	1998	280228.3	311727.5	309193
17	1999	280262.4	311189	311727.5
18	2000	283412.8	310564.8	311189
19	2001	288066.6	306376.4	310564.8
20	2002	290377.5	309039	306376.4

2. TSPプログラムの作成

GiveWin のメニューバーにおいて、「File」－「New」－「Text」とすればプログラムエディターが開く。そこに次のようにプログラムを入力していく。

なお、家計消費支出の変数名を **CE** に変更してある。これは、**C** という変数名が TSP では定数項を表すため、使用できないからである。

```
freq a;
smpl 1983 2002;
read(file='h:ex3-3.csv') Date ce yd yd1;
olsq ce c yd yd1;
ste=@res/@s;
graph date ste;
end;
```

プログラムを入力した後で、メニューバーから「File」－「Save As」を選び、ex3-3という名前で保存する。ファイルの種類は TSP file (*.TSP) を選ぶ。

プログラムの実行は「Modules」－「Run TSP」である。

3. ラグつき変数

この例 5 のモデルは $CE = a + b YD + c YD_{-1}$ というように、説明変数に可処分所得の 1 期前の値が含まれている。このような遅れ(ラグ)をとまなう変数のことを、**ラグつき変数**という。

TSP では、プログラムの中でラグつき変数を計算することができる。

手順

- ① ex3-3.csv のファイルの D 列を削除し、**ex3-3b.csv** という名前で、**CSV 形式**で保存する。
- ② ex3-3.tsp のファイルを次のように修正し、「File」－「Save As」を選び、**ex3-3b** という名前で保存する。(変更箇所を太字で示した)

```
freq a;
smpl 1983 2002;
read(file='h:ex3-3b.csv') Date ce yd;
print ce yd yd(-1);
olsq ce c yd yd(-1);
ste=@res/@s;
graph date ste;
end;
```

ラグつき変数を用いた場合、1983 年の 1 期前のデータは存在しない。そのため、実行結果においていくつかの **warning** が出る。これを回避したい場合は、**smpl** 文で、データの期間を再度定義しなおす必要がある。

4. 相関係数、偏相関係数

TSP では 2 変量間の相関係数の行列を **corr** というコマンドで表示することができる。一方、偏相関係数は計算して導出する必要がある。

ex3-3.tsp を次のように修正し、**ex3-3c.tsp** という名前で保存してみよう。なお、?で始まる文はコメント行である。これらの行は TSP の操作に直接関係ないが、後で自分のプログラムを見て、何をやったのかを確認する時にすると便利である。

```
freq a;
smpl 1983 2002;
read(file='h:ex3-3.csv') Date ce yd yd1;
?
? 重回帰分析
?
olsq ce c yd yd1;
ste=@res/@s;
graph date ste;
?
? 相関係数行列
?
corr ce yd yd1;
?
? 偏相関係数の計算
?
freq n;
smpl 1 1;
pcorr12 = (@corr(1,2)-@corr(1,3)*@corr(2,3))/(sqrt(1-@corr(1,3)^2)*sqrt(1-@corr(2,3)^2));
pcorr13 = (@corr(1,3)-@corr(1,2)*@corr(2,3))/(sqrt(1-@corr(1,2)^2)*sqrt(1-@corr(2,3)^2));
pcorr23 = (@corr(2,3)-@corr(1,2)*@corr(1,3))/(sqrt(1-@corr(1,2)^2)*sqrt(1-@corr(1,3)^2));
?
? 偏相関係数の行列表示
?
pcorrdiag=1;
mmake pcorrmat1 pcorrdiag pcorr12 pcorr13;
mmake pcorrmat2 pcorr12 pcorrdiag pcorr23;
mmake pcorrmat3 pcorr13 pcorr23 pcorrdiag;
mmake(ver) pcorrmat pcorrmat1 pcorrmat2 pcorrmat3;
print pcorrmat;
end;
```

5. 外部ファイルへの書き出し

分析結果や、加工したデータファイルを外部ファイルへ書き出すことができる。Excel で入力したデータを TSP で分析し、ふたたび Excel ファイルとして出力することも可能である。(これによって両者の良い面が使える。)

例として、相関係数行列と偏相関係数行列を Excel 形式で書き出してみよう。end コマンドの前に次のようなコマンドを挿入してみよう。

```
?  
? 相関係数行列、偏相関係数行列のファイル出力  
?  
write(file='h:ex3-3corr.xls') @corr;  
write(file='h:ex3-3pcorr.xls') pcorrmat;
```

6. 分析結果の保存とWordへ読みこみ

TSP の分析結果はテキストファイルとして保存できるので、それをエディターやワープロソフトを使って読みこむことができる。また、グラフも画像形式(デフォルトでは Enhanced MetaFile 形式)で保存できるので、Word などを読みこむことが可能である。

ここでは、ex3-3c の出力結果と残差プロットを Word ファイルにまとめてみよう。

📖 手順

- ① 出力結果のウインドウをアクティブにして、「File」－「Save As」を選び、**ex3-3c** という名前で保存する。(ここで、デフォルトのファイル名には、パス名がついているので、適当に修正すること。)
- ② 残差プロットのウインドウをアクティブにして、「File」－「Save As」を選び、**残差プロット**という名前で保存する。
- ③ Word を開き、**ex3-3c.out** のファイル呼び出す。ファイルの種類を「すべてのファイル」にしておけば、すべての形式のファイル呼び出すことが可能である。
- ④ グラフファイルは、**挿入タブ**をクリックし、「図」ボタンをクリックして、**残差プロット**のファイルを選べば良い。グラフの大きさは、右クリックで「サイズ」を選べば変更できる。
- ⑤ 不要な部分を削除・修正し、タイトルと氏名、考察を加えればレポートが完成する。

☒ **演習問題** : ex3-3c の出力結果と残差プロットをまとめた Word ファイルに、考察を加えたレポートを完成させよう。